

# ORONER: An Ontology for Representing interOrganizational Networks Resilience. A Building Block Towards Adaptive Network Reconfiguration

Marisnel Olivares Amaro<sup>1</sup>[0000-0002-0923-1660], Mohamed Ayoub Laouni<sup>2</sup>[0009-0007-4603-0060], José-Luis Vilchis-Medina<sup>3</sup>[0000-0001-5724-5920], and Éric Saux<sup>1</sup>[0000-0002-6685-6106]

<sup>1</sup> Naval Academy Research Institute (IRENav EA 3634), BCRM de Brest CC 600, F-29240 Brest cedex 9, France

<sup>2</sup> SKEMA Business School, Université Côte d'Azur (GREDEG), France

<sup>3</sup> ENSTA, Lab-STICC (Robex), CNRS UMR 6285, F-29806 Brest, France

olivaresmar@gmail.com, mohamedayoub.laouni@skema.edu, jose.vilchis@ensta.fr, eric.saux@ecole-navale.fr

## 1 Competencies questions, SPARQL queries and results

The following queries were executed in Protégé 5.6.4 using the “Snap SPARQL Query” tool, which relies on a reasoner (HermiT, Pellet, FaCT++, etc.) to provide answers that take OWL semantics into account. Namespaces which are used in the queries are:

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX event: <http://purl.org/NET/c4dm/event.owl#>
PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX time: <http://www.w3.org/2006/time#>
PREFIX oroner: <http://www.semanticweb.org/2022/9/OrganisationResilienceOntology#>
```

### 1.1 General competencies questions about organizations

**CQ1:** How can we characterize an organization (e.g. hospital\_Quimper)?

```
SELECT ?o1 ?o2 ?m1 ?r1 ?h1 #?h2 ?n1
WHERE {
    ?o1 rdf:type oroner:Organization;
    oroner:hasInteraction ?o2;
    oroner:hasMission ?m1;
    oroner:hasResource ?r1;
    oroner:hasHead ?h1.
    # oroner:isMemberofInterOrganizationNetwork ?n1.
    # ?h2 oroner:hasChief ?h1.
    FILTER (regex(str(?o1), "hospital_Quimper"))
}
ORDER BY ?o1
```

Result of the SPARQL query:

?o1	?o2	?m1	?r1	?h1
oroner:hospital_Quimper	oroner:municipality_Quimper	oroner:caringCovid19PatientsQuimper	oroner:head_Hospital_Quimper	oroner:head_Hospital_Quimper
oroner:hospital_Quimper	oroner:municipality_Quimper	oroner:caringCovid19PatientsQuimper	oroner:maskHospitalQuimper	oroner:head_Hospital_Quimper
oroner:hospital_Quimper	oroner:municipality_Quimper	oroner:caringCovid19PatientsQuimper	oroner:clinicalStaffHospitalQuimper	oroner:head_Hospital_Quimper
oroner:hospital_Quimper	oroner:municipality_Quimper	oroner:caringCovid19PatientsQuimper	oroner:hospitalBedsHospitalQuimper	oroner:head_Hospital_Quimper
oroner:hospital_Quimper	oroner:municipality_Quimper	oroner:caringCovid19PatientsQuimper	oroner:ABHSHospitalQuimper	oroner:head_Hospital_Quimper
oroner:hospital_Quimper	oroner:municipality_Quimper	oroner:caringCovid19PatientsQuimper	oroner:respiratorsHospitalQuimper	oroner:head_Hospital_Quimper

36 results

**CQ2:** Who makes the decisions in an organization?

```
SELECT ?o1 ?h1 #?h2
WHERE {
    ?o1 rdf:type oroner:Organization; oroner:hasHead ?h1.
    #?h2 oroner:hasChief ?h1.
}
```

Result of the SPARQL query:

?o1	?h1
oroner:school_Brest	oroner:head_School_Brest
oroner:school_Guipavas	oroner:head_School_Guipavas
oroner:municipality_Quimper	oroner:head_Municipality_Quimper
oroner:interCo_BMO	oroner:head_interCo_BMO
oroner:hospital_Quimper	oroner:head_Hospital_Quimper
oroner:municipality_Brest	oroner:head_Municipality_Brest

14 results

## 1.2 Competencies questions about resources

CQ3: What are the resources and their quantity held by the organizations?

```
SELECT ?x ?r ?t ?v ?q
WHERE {
    ?x rdf:type oroner:Organization; oroner:hasResource ?r.
    ?r rdf:type ?t.
    {?r oroner:hasValue ?v.}
    UNION
    {?r oroner:hasQuantity ?q}
}
```

Result of the SPARQL query:

?x	?r	?t	?v	?q
oroner:municipality_Quimper	oroner:ABHSMunicipalityQuimper	oroner:Resource	96	
oroner:municipality_Quimper	oroner:ABHSMunicipalityQuimper	oroner:AlcoholBasedHandSanitizer	96	
oroner:municipality_Quimper	oroner:ABHSMunicipalityQuimper	owl:Thing	96	
oroner:municipality_Quimper	oroner:policeOfficersMunicipalityQ...	oroner:HumanResource	23	
oroner:municipality_Quimper	oroner:policeOfficersMunicipalityQ...	oroner:Resource	23	
oroner:municipality_Quimper	oroner:policeOfficersMunicipalityQ...	oroner:NonMaterialResource	23	

193 results

CQ4: What are the capabilities possessed by a human agent in an organization and its relationships with other human agents?

```
SELECT ?p1 ?hc1 ?p2
WHERE {
    ?p1 rdf:type oroner:HumanResource.
    {?p1 oroner:hasHumanCapability ?hc1.}
    UNION
    {?p1 oroner:hasRelationWith ?p2.}
}
```

Result of the SPARQL query:

?p1	?hc1	?p2
oroner:ChiefOrga1	oroner:WaitAndSee	
oroner:ChiefOrga1		oroner:ChiefOrga3
oroner:ChiefOrga2	oroner:Proactive	
oroner:ChiefOrga3	oroner:WaitAndSee	
oroner:ChiefOrga4	oroner:Proactive	

5 results

**CQ5: Which chiefs are boundary spanners or have a “Participative” leadership style or “Proactive” human skills?**

```
SELECT ?p1 ?t1 ?ls1 ?hc1
WHERE {
    ?p1 rdf:type oroner:Chief; rdf:type ?t1.
    {
        ?p1 oroner:hasLeadershipStyle ?ls1; oroner:hasHumanCapability ?hc1.
        FILTER (regex(str(?ls1), "Participative") || regex(str(?hc1), "Proactive"))
    }
    UNION
    {
        ?p1 rdf:type oroner:BoundarySpanner.
    }
    FILTER (regex(str(?t1), "Chief") || regex(str(?t1), "BoundarySpanner"))
}
```

Result of the SPARQL query:

?p1	?t1	?ls1	?hc1
oroner:head_School_Brest	oroner:Chief	oroner:Participative	oroner:WaitAndSee
oroner:head_Prefecture_Finistere	oroner:Chief	oroner:Directive	oroner:Proactive
oroner:head_Municipality_Brest	oroner:Chief	oroner:Participative	oroner:Proactive
oroner:head_Hospital_Brest	oroner:Chief		
oroner:head_Hospital_Brest	oroner:BoundarySpanner		

5 results

### 1.3 Competencies questions about missions

**CQ6: What are the missions of organizations by type?**

```
SELECT ?x ?m ?t WHERE {
    ?x rdf:type oroner:Organization.
    ?x oroner:hasMission ?m.
    ?m rdf:type ?t.
}
GROUP BY ?x ?m ?t
ORDER BY ?x ?m ?t
```

Result of the SPARQL query:

?x	?m	?t
oroner:hospital_Brest	oroner:caringAllPatientsBrest	oroner:CaringPatients
oroner:hospital_Brest	oroner:caringAllPatientsBrest	oroner:CitizenMission
oroner:hospital_Brest	oroner:caringAllPatientsBrest	oroner:Mission
oroner:hospital_Brest	oroner:caringAllPatientsBrest	oroner:SocialMission
oroner:hospital_Brest	oroner:caringAllPatientsBrest	owl:Thing
oroner:hospital_Brest	oroner:caringCovid19PatientsBrest	oroner:CaringCovid19Patients

125 results

**CQ6.1: Which organizations are treating Covid 19 patients?**

```
SELECT ?x ?m ?t
WHERE {
    ?x rdf:type oroner:Organization.
    ?x oroner:hasMission ?m.
    ?m rdf:type ?t.
    FILTER (regex(str(?t), "CaringCovid"))
}
```

Result of the SPARQL query:

?x	?m	?t
oroner:hospital_Quimper	oroner:caringCovid19PatientsQuimper	oroner:CaringCovid19Patients
oroner:hospital_Brest	oroner:caringCovid19PatientsBrest	oroner:CaringCovid19Patients

2 results

#### 1.4 Competencies questions about crisis

**CQ7: How can we characterize (factors, impacts, temporal interval, spatial extension, etc.) a crisis?**

```
SELECT ?c1 ?f1 ?d1 ?o1 ?p1 ?geo1 ?hgeo1 ?t1 ?tistart ?ht1start ?tiend ?ht1end
WHERE {
    ?c1 rdf:type oroner:Crisis; event:factor ?f1; event:product ?d1;
    oroner:hasImpactOn ?o1; event:place ?p1; event:time ?t1.
    ?p1 geo:hasGeometry ?geo1.
    ?geo1 geo:asWKT ?hgeo1.
    ?t1 time:hasBeginning ?tistart; time:hasEnd ?tiend.
    ?tistart time:inXSDDate ?ht1start.
    ?tiend time:inXSDDate ?ht1end.
}
```

Result of the SPARQL query:

?c1	?f1	?d1	?o1	?p1	?geo1	?hgeo1	?t1	?tistart	?ht1start	?tiend	?ht1end
oroner:Covid19_Step1	oroner:Factor1	oroner:Thing1	oroner:Orga1	oroner:Place1	oroner:Place1Geo POLYGON((-4.526367 44.01178, ...)	oroner:Interval1	oroner:T1Start	2021-03-17T00:00:00Z	oroner:T1End	2021-05-03T00:00:00Z	

1 results

**CQ8: What are the damages caused by a crisis?**

```
SELECT ?c1 ?d1 ?t1
WHERE {
    ?c1 rdf:type oroner:Crisis; event:product ?d1.
    ?d1 rdf:type ?t1.
}
```

Result of the SPARQL query:

?c1	?d1	?t1
oroner:covid19CrisisMunicipalityBrest	oroner:FunctionalDamage1	oroner:Damage
oroner:covid19CrisisMunicipalityBrest	oroner:FunctionalDamage1	oroner:FunctionalDamage
oroner:covid19CrisisMunicipalityBrest	oroner:FunctionalDamage1	owl:Thing
oroner:Covid19_Step1	oroner:Thing1	oroner:Damage
oroner:Covid19_Step1	oroner:Thing1	owl:Thing

5 results

**CQ9: What is the crisis management status of an organization before, during and after the crisis?**

```
SELECT DISTINCT ?o1 ?cio1 ?ci ?val_bcs1 ?val_dcs1 ?val_acs1
WHERE {
    ?o1 rdf:type oroner:Organization; oroner:isImpactedBy ?cio1.
    ?cio1 oroner:belongsToCrisis ?ci.
    {?cio1 oroner:hasBeforeCrisisManagementStatus ?val_bcs1.}
    UNION
    {?cio1 oroner:hasDuringCrisisManagementStatus ?val_dcs1.}
    UNION
    {?cio1 oroner:hasAfterCrisisManagementStatus ?val_acs1.}
}
```

Result of the SPARQL query:

?o1	?cio1	?ci	?val_bcs1	?val_dcs1	?val_acs1
oroner:municipality_Brest	oroner:covid19CrisisMunicipalityBrest	oroner:Covid19_Step1	oroner:NonPrepared		

1 results

#### 1.5 Competencies questions about interactions

**CQ10: What are the possible types of interactions between organizations?**

```
SELECT ?p1 ?p2 WHERE {
    ?p1 rdf:type owl:ObjectProperty.
    ?p2 rdf:type owl:ObjectProperty.
    ?p1 rdfs:subPropertyOf ?p2.
    FILTER (regex(str(?p2), "hasInteraction"))
}
```

Result of the SPARQL query:

?p1	?p2
oroner:collaborate	oroner:hasInteraction
oroner:cooperate	oroner:hasInteraction
oroner:coordinate	oroner:hasInteraction
oroner:giveAutonomy	oroner:hasInteraction
oroner:giveDelegation	oroner:hasInteraction
oroner:hasControl	oroner:hasInteraction
13 results	

### CQ11: Which organizations are in interaction in an inter-organizational network?

```
SELECT ?n ?o1 ?o2 ?i1 ?t1
WHERE
{
    ?n rdf:type oroner:InterOrganizationNetwork;
    oroner:hasOrganizationMember ?o1; oroner:hasOrganizationMember ?o2.
    ?i1 oroner:hasParticipant ?o1; oroner:hasParticipant ?o2 ;
    oroner:hasType ?t1.
    FILTER (str(?o1)!=str(?o2))
}
```

Result of the SPARQL query:

?n	?o1	?o2	?i1	?t1
oroner:InterOrganizationNetwork1	oroner:Orga2	oroner:Organization3	oroner:Interaction2	SharingResource^xsd:string
oroner:InterOrganizationNetwork1	oroner:Organization3	oroner:Orga2	oroner:Interaction2	SharingResource^xsd:string
2 results				

### CQ12: Which organizations interacted in the past?

```
SELECT ?o1 ?interaction1 ?type1 ?interval ?intervalstart ?xsdTime1
?intervalend ?xsdTime2
WHERE
{
    ?interaction1 oroner:hasParticipant ?o1.
    ?interaction1 time:hasTime ?interval; oroner:hasType ?type1.
    ?interval time:hasBeginning ?intervalstart ; time:hasEnd ?intervalend.
    ?intervalstart time:inXSDDate ?xsdTime1.
    ?intervalend time:inXSDDate ?xsdTime2.
}
```

Result of the SPARQL query:

?o1	?interaction1	?type1	?interval	?intervalstart	?xsdTime1	?intervalend	?xsdTime2
oroner:Orga1	oroner:Interaction1	SharingResource^xsd:string	oroner:Interval1	oroner:T1Start	2021-03-17^xsd:date	oroner:T1End	2021-05-03^xsd:date
oroner:Orga2	oroner:Interaction1	SharingResource^xsd:string	oroner:Interval1	oroner:T1Start	2021-03-17^xsd:date	oroner:T1End	2021-05-03^xsd:date
2 results							

### CQ13: Which chiefs are members of the same meta-organization?

```
SELECT ?m1 ?x1 ?o1 ?x2 ?o2
WHERE {
    ?x1 oroner:isHeadOf ?o1.
    ?x2 oroner:isHeadOf ?o2.
    ?x1 oroner:isHumanResourceOfMetaOrganization ?m1.
    ?x2 oroner:isHumanResourceOfMetaOrganization ?m1.
    FILTER (str(?o1)!=str(?o2))
}
```

Result of the SPARQL query:

?m1	?x1	?o1	?x2	?o2
oroner:MedecinsSansFrontieres	oroner:head_Hospital_Brest	oroner:hospital_Brest	oroner:head_Hospital_Quimper	oroner:hospital_Quimper
oroner:MedecinsSansFrontieres	oroner:head_Hospital_Quimper	oroner:hospital_Quimper	oroner:head_Hospital_Brest	oroner:hospital_Brest
oroner:associationMairesDeFrance	oroner:head_Municipality_Brest	oroner:municipality_Brest	oroner:head_Municipality_Quimper	oroner:municipality_Quimper
oroner:associationMairesDeFrance	oroner:head_Municipality_Quimper	oroner:municipality_Quimper	oroner:head_Municipality_Brest	oroner:municipality_Brest

4 results

#### CQ14: Which organizations cooperate in the network?

```
SELECT ?o1 ?o2
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o1 oroner:cooperate ?o2 .
}
```

Result of the SPARQL query:

?o1	?o2
oroner:school_Brest	oroner:municipality_Brest
oroner:school_Brest	oroner:hospital_Brest
oroner:municipality_Quimper	oroner:hospital_Quimper
oroner:municipality_Quimper	oroner:school_Quimper
oroner:interCo_BMO	oroner:municipality_Brest
oroner:hospital_Quimper	oroner:municipality_Quimper

14 results

#### CQ15: Which organizations are in conflict with and who has authority over them?

```
SELECT ?o1 ?o2 ?o3
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o3 rdf:type oroner:Organization.
    ?o1 oroner:isInConflict ?o2 .
    ?o3 oroner:hasVerticalInteraction ?o2 .
}
```

Result of the SPARQL query:

?o1	?o2	?o3
oroner:interCo_BMO	oroner:municipality_Guipavas	oroner:prefecture_Finistere
oroner:municipality_Brest	oroner:municipality_Guipavas	oroner:prefecture_Finistere
oroner:municipality_Guipavas	oroner:municipality_Brest	oroner:prefecture_Finistere

3 results

#### CQ16: Which organizations are in conflict and which are under their control?

```
SELECT ?o1 ?o2 ?o3
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o3 rdf:type oroner:Organization.
    ?o1 oroner:isInConflict ?o2 .
    ?o2 oroner:hasVerticalInteraction ?o3 .
}
```

Result of the SPARQL query:

?o1	?o2	?o3
oroner:prefecture_Morbihan	oroner:prefecture_Finistere	oroner:municipality_Quimper
oroner:prefecture_Morbihan	oroner:prefecture_Finistere	oroner:municipality_Brest
oroner:prefecture_Morbihan	oroner:prefecture_Finistere	oroner:municipality_Guipavas
3 results		

**CQ17: Which organizations have horizontal interactions and have care missions?**

```
SELECT ?o1 ?o2 ?m1 ?m2
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o1 oroner:hasHorizontalInteraction ?o2.
    ?o1 oroner:hasMission ?m1.
    ?o2 oroner:hasMission ?m2.
    FILTER(CONTAINS(xsd:string(?m1),"caring")
    && CONTAINS(xsd:string(?m2),"caring"))
}
```

Result of the SPARQL query:

?o1	?o2	?m1	?m2
oroner:hospital_Quimper	oroner:hospital_Brest	oroner:caringCovid19PatientsQuimper	oroner:caringCovid19PatientsBrest
oroner:hospital_Quimper	oroner:hospital_Brest	oroner:caringCovid19PatientsQuimper	oroner:caringAllPatientsBrest
oroner:hospital_Quimper	oroner:hospital_Brest	oroner:caringAllPatientsQuimper	oroner:caringCovid19PatientsBrest
oroner:hospital_Quimper	oroner:hospital_Brest	oroner:caringAllPatientsQuimper	oroner:caringAllPatientsBrest
oroner:hospital_Brest	oroner:hospital_Quimper	oroner:caringCovid19PatientsBrest	oroner:caringCovid19PatientsQuimper
oroner:hospital_Brest	oroner:hospital_Quimper	oroner:caringCovid19PatientsBrest	oroner:caringAllPatientsQuimper

8 results

**CQ18: Which organizations have horizontal interactions and share the same missions?**

```
SELECT ?o1 ?o2 ?m1 ?m2
WHERE {
    ?o1 rdf:type oroner:Organization.
    ?o2 rdf:type oroner:Organization.
    ?o1 oroner:hasHorizontalInteraction ?o2.
    ?o1 oroner:hasMission ?m1.
    ?o2 oroner:hasMission ?m2.
    FILTER(SUBSTR(xsd:string(?m1), 64,8) = SUBSTR(xsd:string(?m2), 64,8))
}
```

Result of the SPARQL query:

?o1	?o2	?m1	?m2
oroner:school_Brest	oroner:school_Guipavas	oroner:hostingChildrenClinicalStaffBrest	oroner:hostingChildrenClinicalStaffGuipav...
oroner:school_Brest	oroner:school_Guipavas	oroner:educatingChildrenBrest	oroner:educatingChildrenGuipavas
oroner:school_Brest	oroner:school_Quimper	oroner:hostingChildrenClinicalStaffBrest	oroner:hostingChildrenClinicalStaffQuimper
oroner:school_Brest	oroner:school_Quimper	oroner:educatingChildrenBrest	oroner:educatingChildrenQuimper
oroner:school_Guipavas	oroner:school_Brest	oroner:hostingChildrenClinicalStaffGuipav...	oroner:hostingChildrenClinicalStaffBrest
oroner:school_Guipavas	oroner:school_Brest	oroner:educatingChildrenGuipavas	oroner:educatingChildrenBrest

46 results

## 2 Rules

The purpose of the implemented rules is to determine whether resource sharing is possible between two organizations. Below, we present some of the rules implemented in ORONER using the “SWRL Tab” plugin in Protégé. The inferred axioms are provided after each rule.

**R1: If ?x and ?y are organizations that interact horizontally in a positive way, then they can share their resources.**

```
orga:Organization(?x) ^ orga:Organization(?y)
^ orga:hasPositiveHorizontalInteraction(?x, ?y)
-> orga:PossibleResourceSharing(?x, ?y)
```

Inferred axioms:

```
school_Quimper PossibleResourceSharing school_Brest
school_Brest PossibleResourceSharing municipality_Brest
hospital_Brest PossibleResourceSharing municipality_Brest
school_Guipavas PossibleResourceSharing school_Brest
hospital_Brest PossibleResourceSharing school_Brest
hospital_Brest PossibleResourceSharing hospital_Quimper
interCo_BMO PossibleResourceSharing municipality_Brest
hospital_Quimper PossibleResourceSharing hospital_Brest
school_Brest PossibleResourceSharing school_Guipavas
municipality_Brest PossibleResourceSharing school_Brest
school_Guipavas PossibleResourceSharing municipality_Guipavas
hospital_Quimper PossibleResourceSharing municipality_Quimper
school_Quimper PossibleResourceSharing municipality_Quimper
school_Brest PossibleResourceSharing hospital_Brest
municipality_Guipavas PossibleResourceSharing school_Guipavas
municipality_Quimper PossibleResourceSharing school_Quimper
municipality_Quimper PossibleResourceSharing hospital_Quimper
municipality_Brest PossibleResourceSharing interCo_BMO
school_Brest PossibleResourceSharing school_Quimper
school_Quimper PossibleResourceSharing hospital_Quimper
hospital_Quimper PossibleResourceSharing school_Quimper
municipality_Brest PossibleResourceSharing hospital_Brest
```

**R2: If organizations ?x and ?y interact horizontally in a positive way and have the same mission (e.g. CaringCovid19Patients), then they can share their resources.**

```
orga:Organization(?x) ^ orga:Organization(?y) ^ orga:hasPositiveHorizontalInteraction(?x, ?y)
^ orga:hasMission(?x, ?mx) ^ orga:CaringCovid19Patients(?mx) ^ orga:hasMission(?y, ?my)
^ orga:CaringCovid19Patients(?my) -> orga:PossibleResourceSharing(?x, ?y)
```

Inferred axioms:

```
hospital_Quimper PossibleResourceSharing hospital_Brest
hospital_Brest PossibleResourceSharing hospital_Quimper
```

**R3: If ?z is a meta-organization which is composed of different organizations ?x and ?y, then resources can be shared between them.**

```
orga:MetaOrganization(?z) ^ orga:isComposedOfOrganization(?z, ?x)
^ orga:isComposedOfOrganization(?z, ?y) ^ differentFrom(?x, ?y)
-> orga:PossibleResourceSharing(?x, ?y)
```

Inferred axioms:

```
associationMairesDeFrance29 PossibleResourceSharing associationMairesDeFrance13
associationMairesDeFrance13 PossibleResourceSharing associationMairesDeFrance29
```

**R4: If ?x1 is the head of an organization ?y1 and a member of a meta-organization ?z and if ?x2 is the head of an organization ?y2 and a member of the same meta-organization ?z, then organizations ?y1 and ?y2 can share their resources.**

```
orga:isHumanResourceOfMetaOrganization(?x1, ?z) ^ orga:isHumanResourceOfMetaOrganization(?x2, ?z)
^ orga:isHeadOf(?x1, ?y1) ^ orga:isHeadOf(?x2, ?y2) ^ differentFrom(?y1, ?y2)
-> orga:PossibleResourceSharing(?y1, ?y2)
```

Inferred axioms:

```
municipality_Quimper PossibleResourceSharing municipality_Brest
hospital_Brest PossibleResourceSharing hospital_Quimper
municipality_Brest PossibleResourceSharing municipality_Quimper
hospital_Quimper PossibleResourceSharing hospital_Brest
```

**R5:** If  $?x$  is the head of an organization  $?y$  and a member of a meta-organization  $?z$  to which an organization  $?w$  belongs, then organizations  $?y$  and  $?w$  can share their resources.

```
orga:isHumanResourceOfMetaOrganization(?x, ?z)
^ orga:isComposedOfOrganization(?z, ?w) ^ orga:isHeadOf(?x, ?y)
^ differentFrom(?y, ?w) -> orga:PossibleResourceSharing(?y, ?w)
```

Inferred axioms:

```
municipality_Brest PossibleResourceSharing associationMairesDeFrance29
municipality_Quimper PossibleResourceSharing associationMairesDeFrance29
municipality_Brest PossibleResourceSharing associationMairesDeFrance13
municipality_Quimper PossibleResourceSharing associationMairesDeFrance13
associationMairesDeFrance13 PossibleResourceSharing municipality_Quimper
associationMairesDeFrance29 PossibleResourceSharing municipality_Brest
associationMairesDeFrance29 PossibleResourceSharing municipality_Quimper
associationMairesDeFrance13 PossibleResourceSharing municipality_Brest
```

**R6:** If  $?x_1$  is a top level management of an organization  $?y_1$  and a member of a meta-organization  $?z$  and if  $?x_2$  is a top level management of an organization  $?y_2$  and a member of the same meta-organization  $?z$ , then organizations  $?y_1$  and  $?y_2$  can share their resources.

```
orga:isHumanResourceOfMetaOrganization(?x1, ?z) ^ orga:isHumanResourceOfMetaOrganization(?x2, ?z)
^ orga:TopLevelManagement(?x1) ^ orga:isHumanResourceOf(?x1, ?y1)
^ orga:TopLevelManagement(?x2) ^ orga:isHumanResourceOf(?x2, ?y2) ^ differentFrom(?y1, ?y2)
-> orga:PossibleResourceSharing(?y1, ?y2)
```

Inferred axioms:

```
None % because there is no TopLevelManagement who is a human member of a metaorganization in the use case
```

**R7:** If  $?x$  is a top level management of an organization  $?y$  and a member of a meta-organization  $?z$  to which an organization  $?w$  belongs, then organizations  $?y$  and  $?w$  can share their resources.

```
orga:isHumanResourceOfMetaOrganization(?x, ?z) ^ orga:TopLevelManagement(?x)
^ orga:isHumanResourceOf(?x, ?y) ^ orga:isComposedOfOrganization(?z, ?w)
^ differentFrom(?y, ?w) -> orga:PossibleResourceSharing(?y, ?w)
```

Inferred axioms:

```
None % because there is no TopLevelManagement who is a human member of a metaorganization in the use case
```

**R8:** If  $?x$  is a boundary spanner between two different organizations, then they can share their resources.

```
orga:BoundarySpanner(?x) ^ orga:isHumanResourceOf(?x, ?y)
^ orga:isHumanResourceOf(?x, ?z) ^ differentFrom(?y, ?z)
-> orga:PossibleResourceSharing(?y, ?z)
```

Inferred axioms:

```
Orga2 PossibleResourceSharing Orga1
Orga1 PossibleResourceSharing Orga2
municipality_Brest PossibleResourceSharing hospital_Brest
hospital_Brest PossibleResourceSharing municipality_Brest
```